Furman ratio without Cobb-Douglas

For discrete changes in the capital-income tax rate.

TheoryGuru applications

Background

Obama administration economists Furman and Summers claimed that only a fraction of the revenue loss from a corporate tax cut benefits labor. But the standard supply and demand model, which for these purposes is a generalization of long run behavior in the neoclassical growth model, says the opposite.

Here we prove that by machine, without assuming any functional form for the aggregate production function. k denotes the aggregate capital stock, $f[k]$ aggregate output gross of depreciation (the aggregate quantity of labor is fixed), and τ the capital-income tax rate.

Setup

In[ ]:=
```
Get@"http://economicreasoning.com"
```

Proof & Logic Tools 6.3

Type ERCommands for a list of commands in the package.

Introduction to Automated Economic Reasoning		
Tutorials:	Entering calculus	General Mathematica tips
Get started	Load extras	Browse examples

Definitions

In[3]:=
```
laborincome[k_] = f[k] - f'[k] k
```

Out[3]= $f[k] - k f'[k]$

```
In[4]:= lrcapitalequilibrium[τ_, k_] := (* willingness to pay for capital *)
   (1 - τ) f'[k] == ρ + δ (* LR willingness to supply it *)

In[5]:= signconditions =
   {δ > 0, ρ > 0, k1 > 0, k2 > 0, 0 ≤ τ1 < τ2 < 1, SameSign[f'[k2] - f'[k1], k1 - k2],
    (* concave production *) f'[k1] (k1 - k2) < f[k1] - f[k2] < f'[k2] (k1 - k2) ∨ k1 == k2,
    SameSign[laborincome[k2] - laborincome[k1], k2 - k1],
    SameSign[f[k2] - f[k1], k2 - k1]};

In[6]:= revenue[τ_, k_] := τ (f'[k] - δ) k
```

In[7]:= furmanratio := $\frac{\texttt{laborincome}[k_2] - \texttt{laborincome}[k_1]}{\texttt{revenue}[\tau_1, k_1] - \texttt{revenue}[\tau_2, k_2]}$

Interesting but not necessary assumptions

```
In[8]:= elasticcapitaldemand = (k2 f'[k2] - k1 f'[k1]) (k2 - k1) ≥ 0;

In[9]:= wrongsideoflaffercurve = (revenue[τ2, k2] - revenue[τ1, k1]) (τ2 - τ1) ≤ 0;
```

Results

Taxation reduces the stock capital and the amount of labor income

```
In[ ]:= TheoryGuru[{lrcapitalequilibrium[τ1, k1], lrcapitalequilibrium[τ2, k2],
    Most@signconditions},
   k2 < k1 ∧ laborincome[k1] > laborincome[k2]]

Out[ ]= True
```

Taxation reduces labor income more than it increases revenue

```
In[ ]:= TheoryGuru[{lrcapitalequilibrium[τ1, k1], lrcapitalequilibrium[τ2, k2],
    Most@signconditions},
   revenue[τ1, k1] + laborincome[k1] > revenue[τ2, k2] + laborincome[k2]]

Out[ ]= True
```

Either the Furman ratio exceeds one or the tax is reducing revenue

In[10]:=
```
TheoryGuru[{lrcapitalequilibrium[τ1, k1], lrcapitalequilibrium[τ2, k2],
  Most@Most@signconditions},

 furmanratio > 1
  ∨
  wrongsideoflaffercurve]
```

Out[10]=
```
True
```

In[ ]:=
```
TheoryGuru[{lrcapitalequilibrium[τ1, k1], lrcapitalequilibrium[τ2, k2],
  Most@Most@signconditions},

 furmanratio > 1
  ∨
  furmanratio < 0]
```

Out[ ]=
```
True
```